
power twocorrelations — Power analysis for a two-sample correlations test

Description Quick start Menu Syntax
Options Remarks and examples Stored results Methods and formulas
References Also see

Description
power twocorrelations computes sample size, power, or the experimental-group correlation for

a two-sample correlations test. By default, it computes sample size for given power and the values of

the control-group and experimental-group correlations. Alternatively, it can compute power for given

sample size and values of the control-group and experimental-group correlations or the experimental-

group correlation for given sample size, power, and the control-group correlation. Also see [PSS-2] power

for a general introduction to the power command using hypothesis tests.

Quick start
Sample size for a test of 𝐻0∶ 𝜌1 = 𝜌2 versus 𝐻𝑎∶ 𝜌1 ≠ 𝜌2 when control-group correlation 𝑟1 = 0.4 and

experimental-group correlation 𝑟2 = 0.1 with default power of 0.8 and significance level 𝛼 = 0.05

power twocorrelations .4 .1

Same as above, for 𝑟2 equal to −0.05, 0, 0.05, and 0.1

power twocorrelations .4 (-.05(.05).1)

Same as above, but for a one-sided test

power twocorrelations .4 (-.05(.05).1), onesided

Specify 𝑟1 and difference 𝑟2 − 𝑟1 = −0.3

power twocorrelations .4, diff(-.3)

Same as above, but specify a sample size of 200 for group 1 and compute sample size for group 2

power twocorrelations 0.4, diff(-0.3) n1(200) compute(N2)

Power for 𝑟1 = 0.4 and 𝑟2 = −0.15 with a sample size of 100 and default 𝛼 = 0.05

power twocorrelations .4 -.15, n(100)

Same as above, but specify sample size of 50 and 65 for groups 1 and 2, respectively

power twocorrelations .4 -.15 , n1(50) n2(65)

Power for sample sizes of 60, 70, 80, 90, and 100

power twocorrelations .4 -.15, n(60(10)100)

Same as above, but display results in a graph of power versus sample size

power twocorrelations .4 -.15, n(60(10)100) graph

Effect size and experimental-group correlation for 𝑟1 = 0.4 with sample size of 150 and power of 0.85

power twocorrelations .4, n(150) power(.85)

Menu
Statistics > Power, precision, and sample size

Syntax
Compute sample size

power twocorrelations 𝑟1 𝑟2 [ , power(numlist) options ]

Compute power

power twocorrelations 𝑟1 𝑟2 , n(numlist) [ options ]

Compute effect size and experimental-group correlation

power twocorrelations 𝑟1 , n(numlist) power(numlist) [ options ]

where 𝑟1 is the correlation in the control (reference) group and 𝑟2 is the correlation in the experimental

(comparison) group. 𝑟1 and 𝑟2 may each be specified either as one number or as a list of values in

parentheses (see [U] 11.1.8 numlist).

https://www.stata.com/manuals/u11.pdf#u11.1.8numlist
https://www.stata.com/manuals/pss-2powertwocorrelations.pdf#pss-2powertwocorrelationsSyntaxoptions
https://www.stata.com/manuals/u11.pdf#u11.1.8numlist
https://www.stata.com/manuals/pss-2powertwocorrelations.pdf#pss-2powertwocorrelationsSyntaxoptions
https://www.stata.com/manuals/u11.pdf#u11.1.8numlist
https://www.stata.com/manuals/u11.pdf#u11.1.8numlist
https://www.stata.com/manuals/pss-2powertwocorrelations.pdf#pss-2powertwocorrelationsSyntaxoptions
https://www.stata.com/manuals/u11.pdf#u11.1.8numlist

options Description

Main
∗ alpha(numlist) significance level; default is alpha(0.05)
∗ power(numlist) power; default is power(0.8)
∗ beta(numlist) probability of type II error; default is beta(0.2)
∗ n(numlist) total sample size; required to compute power or effect size
∗ n1(numlist) sample size of the control group
∗ n2(numlist) sample size of the experimental group
∗ nratio(numlist) ratio of sample sizes, N2/N1; default is nratio(1), meaning

equal group sizes

compute(N1 | N2) solve for N1 given N2 or for N2 given N1
nfractional allow fractional sample sizes

∗ diff(numlist) difference between the experimental-group and
control-group correlations, 𝑟2 − 𝑟1; specify instead of the
experimental-group correlation 𝑟2

direction(upper|lower) direction of the effect for effect-size determination; default is
direction(upper), which means that the postulated value
of the parameter is larger than the hypothesized value

onesided one-sided test; default is two sided

parallel treat number lists in starred options or in command arguments as
parallel when multiple values per option or argument are
specified (do not enumerate all possible combinations of values)

Table

[ no ]table[ (tablespec) ] suppress table or display results as a table;
see [PSS-2] power, table

saving(filename [ , replace ]) save the table data to filename; use replace to overwrite
existing filename

Graph

graph[ (graphopts) ] graph results; see [PSS-2] power, graph

Iteration

init(#) initial value for sample sizes or experimental-group correlation

iterate(#) maximum number of iterations; default is iterate(500)
tolerance(#) parameter tolerance; default is tolerance(1e-12)
ftolerance(#) function tolerance; default is ftolerance(1e-12)
[ no ]log suppress or display iteration log

[ no ]dots suppress or display iterations as dots

notitle suppress the title

∗Specifying a list of values in at least two starred options, or at least two command arguments, or at least one
starred option and one argument results in computations for all possible combinations of the values; see

[U] 11.1.8 numlist. Also see the parallel option.

collect is allowed; see [U] 11.1.10 Prefix commands.

notitle does not appear in the dialog box.

https://www.stata.com/manuals/u11.pdf#u11.1.8numlist
https://www.stata.com/manuals/u11.pdf#u11.1.8numlist
https://www.stata.com/manuals/u11.pdf#u11.1.8numlist
https://www.stata.com/manuals/u11.pdf#u11.1.8numlist
https://www.stata.com/manuals/u11.pdf#u11.1.8numlist
https://www.stata.com/manuals/u11.pdf#u11.1.8numlist
https://www.stata.com/manuals/u11.pdf#u11.1.8numlist
https://www.stata.com/manuals/u11.pdf#u11.1.8numlist
https://www.stata.com/manuals/pss-2.pdf#pss-2powertwocorrelationsSyntaxtablespec
https://www.stata.com/manuals/pss-2powertable.pdf#pss-2power,table
https://www.stata.com/manuals/u11.pdf#u11.6Filenamingconventions
https://www.stata.com/manuals/pss-2powergraph.pdf#pss-2power,graphSyntaxgraphopts
https://www.stata.com/manuals/pss-2powergraph.pdf#pss-2power,graph
https://www.stata.com/manuals/u11.pdf#u11.1.8numlist
https://www.stata.com/manuals/u11.pdf#u11.1.10Prefixcommands

where tablespec is

column[ :label ] [ column[ :label ] [ . . . ] ] [ , tableopts ]

column is one of the columns defined below, and label is a column label (may contain quotes and com-

pound quotes).

column Description Symbol

alpha significance level 𝛼
power power 1 − 𝛽
beta type-II-error probability 𝛽
N total number of subjects 𝑁
N1 number of subjects in the control group 𝑁1
N2 number of subjects in the experimental group 𝑁2
nratio ratio of sample sizes, experimental to control 𝑁2/𝑁1
delta effect size 𝛿
r1 control-group correlation 𝜌1
r2 experimental-group correlation 𝜌2
diff difference between the experimental-group correlation 𝜌2 − 𝜌1

and the control-group correlation

target target parameter; synonym for r2
all display all supported columns

Column beta is shown in the default table in place of column power if specified.

Columns diff and nratio are shown in the default table if specified.

Options

� � �
Main �

alpha(), power(), beta(), n(), n1(), n2(), nratio(), compute(), nfractional; see

[PSS-2] power.

diff(numlist) specifies the difference between the experimental-group correlation and the control-

group correlation, 𝑟2 − 𝑟1. You can specify either the experimental-group correlation 𝑟2 as a com-

mand argument or the difference between the two correlations in diff(). If you specify diff(#),
the experimental-group correlation is computed as 𝑟2 = 𝑟1 + #. This option is not allowed with the

effect-size determination.

direction(), onesided, parallel; see [PSS-2] power.

� � �
Table �

table, table(), notable; see [PSS-2] power, table.

saving(); see [PSS-2] power.

� � �
Graph �

graph, graph(); see [PSS-2] power, graph. Also see the column table for a list of symbols used by the

graphs.

https://www.stata.com/manuals/pss-2.pdf#pss-2powertwocorrelationsSyntaxcolumn
https://www.stata.com/manuals/pss-2powertable.pdf#pss-2power,tableSyntaxtableopts
https://www.stata.com/manuals/pss-2.pdf#pss-2powertwocorrelationsSyntaxcolumn
https://www.stata.com/manuals/pss-2power.pdf#pss-2power
https://www.stata.com/manuals/u11.pdf#u11.1.8numlist
https://www.stata.com/manuals/pss-2power.pdf#pss-2power
https://www.stata.com/manuals/pss-2powertable.pdf#pss-2power,table
https://www.stata.com/manuals/pss-2power.pdf#pss-2power
https://www.stata.com/manuals/pss-2powergraph.pdf#pss-2power,graph
https://www.stata.com/manuals/pss-2.pdf#pss-2powertwocorrelationsSyntaxcolumn

� � �
Iteration �

init(#) specifies the initial value for the estimated parameter. For sample-size determination, the esti-

mated parameter is either the control-group size 𝑛1 or, if compute(N2) is specified, the experimental-
group size 𝑛2. For the effect-size determination, the estimated parameter is the experimental-group

correlation 𝑟2. The default initial values for a two-sided test are obtained as a closed-form solution

for the corresponding one-sided test with the significance level 𝛼/2.
iterate(), tolerance(), ftolerance(), log, nolog, dots, nodots; see [PSS-2] power.

The following option is available with power twocorrelations but is not shown in the dialog box:

notitle; see [PSS-2] power.

Remarks and examples
Remarks are presented under the following headings:

Introduction
Using power twocorrelations
Computing sample size
Computing power
Computing effect size and experimental-group correlation
Testing a hypothesis about two independent correlations

This entry describes the power twocorrelations command and the methodology for power and

sample-size analysis for a two-sample correlations test. See [PSS-2] Intro (power) for a general intro-

duction to power and sample-size analysis and [PSS-2] power for a general introduction to the power
command using hypothesis tests.

Introduction
There are many examples of studies where a researcher may want to compare two correlations. A

winemaker may want to test the null hypothesis that the correlation between fermentation time and alco-

hol level is the same for pinot noir grapes and merlot grapes. An education researcher may want to test

the null hypothesis that the correlation of verbal and math SAT scores is the same for males and females.

Or a genetics researcher may want to test the null hypothesis that the correlation of the cholesterol levels

in identical twins raised together is equal to the correlation of the cholesterol levels in identical twins

raised apart.

Correlation analysis emanates from studies quantifying dependence between random variables. The

correlation coefficient 𝜌 is commonly used to measure the strength of such dependence. We consider

Pearson’s correlation obtained by standardizing the covariance between two random variables. As a

result, the correlation coefficient has no units and ranges between −1 and 1.

This entry describes power and sample-size analysis for the inference about two population correlation

coefficients performed using hypothesis testing. Specifically, we consider the null hypothesis 𝐻0∶ 𝜌2 =
𝜌1 versus the two-sided alternative hypothesis𝐻𝑎∶ 𝜌2 ≠ 𝜌1, the upper one-sided alternative𝐻𝑎∶ 𝜌2 > 𝜌1,

or the lower one-sided alternative 𝐻𝑎∶ 𝜌2 < 𝜌1.

One common approach in testing hypotheses concerning the correlation parameter is to use an inverse

hyperbolic tangent transformation, tanh−1(𝑥) = 0.5 ln(1 + 𝑥)/ ln(1 − 𝑥), also known as Fisher’s 𝑧
transformation when applied to the correlation coefficient (Fisher 1915). Specifically, if ̂𝜌 is the sample

correlation coefficient and 𝑛 is the sample size, Fisher (1915) showed that
√

𝑛 − 3 {tanh−1( ̂𝜌) − tanh−1(𝜌)} ∼ 𝑁(0, 1)

https://www.stata.com/manuals/pss-2power.pdf#pss-2power
https://www.stata.com/manuals/pss-2power.pdf#pss-2power
https://www.stata.com/manuals/pss-2intropower.pdf#pss-2Intro(power)
https://www.stata.com/manuals/pss-2power.pdf#pss-2power

for 𝑛 as small as 10, although the approximation tends to perform better for 𝑛 > 25. For a two-sample

correlations test, the null hypothesis 𝐻0∶ 𝜌1 = 𝜌2 is equivalent to 𝐻0∶ tanh−1(𝜌1) = tanh−1(𝜌2). The
latter test is referred to as the two-sample Fisher’s 𝑧 test.

power twocorrelations performs computations based on the asymptotic two-sample Fisher’s

𝑧 test.

Using power twocorrelations
power twocorrelations computes sample size, power, or experimental-group correlation for a two-

sample correlations test. All computations are performed for a two-sided hypothesis test where, by de-

fault, the significance level is set to 0.05. You may change the significance level by specifying the

alpha() option. You can specify the onesided option to request a one-sided test. By default, all com-

putations assume a balanced or equal-allocation design; see [PSS-4]Unbalanced designs for a description

of how to specify an unbalanced design.

To compute the total sample size, you must specify the control- and experimental-group correlations,

𝑟1 and 𝑟2, respectively, and, optionally, the power of the test in the power() option. The default power

is set to 0.8.

Instead of the total sample size, you can compute one of the group sizes given the other one. To

compute the control-group sample size, you must specify the compute(N1) option and the sample size

of the experimental group in the n2() option. Likewise, to compute the experimental-group sample size,

you must specify the compute(N2) option and the sample size of the control group in the n1() option.

To compute power, you must specify the total sample size in the n() option and the control- and

experimental-group correlations, 𝑟1 and 𝑟2, respectively.

Instead of the experimental-group correlation 𝑟2, you may specify the difference 𝑟2 − 𝑟1 between the

experimental-group correlation and the control-group correlation in the diff() option when computing

sample size or power.

To compute effect size, the difference between the experimental-group and the control-group corre-

lation, and the experimental-group correlation, you must specify the total sample size in the n() option,

the power in the power() option, the control-group correlation 𝑟1, and optionally, the direction of the

effect. The direction is upper by default, direction(upper), which means that the experimental-group
correlation is assumed to be larger than the specified control-group value. You can change the direc-

tion to be lower, which means that the experimental-group correlation is assumed to be smaller than the

specified control-group value, by specifying the direction(lower) option.

Instead of the total sample size n(), you can specify individual group sizes in n1() and n2(), or
specify one of the group sizes and nratio()when computing power or effect size. Also see Two samples
in [PSS-4] Unbalanced designs for more details.

In the following sections, we describe the use of power twocorrelations accompanied with exam-

ples for computing sample size, power, and experimental-group correlation.

Computing sample size
To compute sample size, you must specify the control- and experimental-group correlations, 𝑟1 and

𝑟2, respectively, and, optionally, the power of the test in the power() option. A default power of 0.8 is

assumed if power() is not specified.

https://www.stata.com/manuals/pss-4unbalanceddesigns.pdf#pss-4Unbalanceddesigns
https://www.stata.com/manuals/pss-4unbalanceddesigns.pdf#pss-4UnbalanceddesignsRemarksandexamplesTwosamples
https://www.stata.com/manuals/pss-4unbalanceddesigns.pdf#pss-4Unbalanceddesigns

Example 1: Sample size for a two-sample correlations test
Consider a study in which investigators are interested in testingwhether the correlation between height

and weight differs for males and females. The null hypothesis 𝐻0 ∶ 𝜌𝐹 = 𝜌𝑀 is tested against the

alternative hypothesis 𝐻𝑎∶ 𝜌𝐹 ≠ 𝜌𝑀, where 𝜌𝐹 is the correlation between height and weight for females

and 𝜌𝑀 is the correlation between height and weight for males.

Before conducting the study, investigators wish to determine the minimum sample size required to

detect a difference between the correlation of 0.3 for females and a correlation of 0.5 for males with 80%

power using a two-sided 5%-level test. We specify the values 0.3 and 0.5 as the control- and experimental-

group correlations after the command name. We omit options alpha(0.05) and power(0.8) because

the specified values are their defaults. To compute the total sample size assuming equal-group allocation,

we type

. power twocorrelations 0.3 0.5
Performing iteration ...
Estimated sample sizes for a two-sample correlations test
Fisher’s z test
H0: r2 = r1 versus Ha: r2 != r1
Study parameters:

alpha = 0.0500
power = 0.8000
delta = 0.2000

r1 = 0.3000
r2 = 0.5000

Estimated sample sizes:
N = 554

N per group = 277

A total sample of 554 individuals, 277 in each group, must be obtained to detect the difference between

correlations of females and males when the correlations are 0.3 and 0.5, respectively, with 80% power

using a two-sided 5%-level test.

Example 2: Computing one of the group sizes
Continuing with example 1, we will suppose that for some reason, we can enroll only 250 male

subjects in our study. We want to know how many female subjects we need to recruit to maintain the

80% power for detecting the difference as described in example 1. To do so, we use the combination of

compute(N1) and n2():

https://www.stata.com/manuals/pss-2powertwocorrelations.pdf#pss-2powertwocorrelationsRemarksandexamplesex1

. power twocorrelations 0.3 0.5, n2(250) compute(N1)
Performing iteration ...
Estimated sample sizes for a two-sample correlations test
Fisher’s z test
H0: r2 = r1 versus Ha: r2 != r1
Study parameters:

alpha = 0.0500
power = 0.8000
delta = 0.2000

r1 = 0.3000
r2 = 0.5000
N2 = 250

Estimated sample sizes:
N = 559

N1 = 309

We need 309 females for a total sample size of 559 subjects, which is larger than the required total sample

size for the corresponding balanced design from example 1.

Example 3: Unbalanced design
By default, power twocorrelations computes sample size for a balanced or equal-allocation de-

sign. If we know the allocation ratio of subjects between the groups, we can compute the required sample

size for an unbalanced design by specifying the nratio() option.

Continuing with example 1, we will suppose that we anticipate to recruit twice as many males as

females; that is, 𝑛2/𝑛1 = 2. We specify the nratio(2) option to compute the required sample size for

the specified unbalanced design.

. power twocorrelations 0.3 0.5, nratio(2)
Performing iteration ...
Estimated sample sizes for a two-sample correlations test
Fisher’s z test
H0: r2 = r1 versus Ha: r2 != r1
Study parameters:

alpha = 0.0500
power = 0.8000
delta = 0.2000

r1 = 0.3000
r2 = 0.5000

N2/N1 = 2.0000
Estimated sample sizes:

N = 624
N1 = 208
N2 = 416

Also see Two samples in [PSS-4] Unbalanced designs for more examples of unbalanced designs for

two-sample tests.

Computing power
To compute power, you must specify the sample size in the n() option and the control- and the

experimental-group correlations, 𝑟1 and 𝑟2, respectively.

https://www.stata.com/manuals/pss-2powertwocorrelations.pdf#pss-2powertwocorrelationsRemarksandexamplesex1
https://www.stata.com/manuals/pss-2powertwocorrelations.pdf#pss-2powertwocorrelationsRemarksandexamplesex1
https://www.stata.com/manuals/pss-4unbalanceddesigns.pdf#pss-4UnbalanceddesignsRemarksandexamplesTwosamples
https://www.stata.com/manuals/pss-4unbalanceddesigns.pdf#pss-4Unbalanceddesigns

Example 4: Power of a two-sample correlations test
Continuingwith example 1, wewill suppose that we are designing a study and anticipate a total sample

of 500 subjects. To compute the power corresponding to this sample size given the study parameters from

example 1, we specify the sample size in n():
. power twocorrelations 0.3 0.5, n(500)
Estimated power for a two-sample correlations test
Fisher’s z test
H0: r2 = r1 versus Ha: r2 != r1
Study parameters:

alpha = 0.0500
N = 500

N per group = 250
delta = 0.2000

r1 = 0.3000
r2 = 0.5000

Estimated power:
power = 0.7595

With a smaller sample of 500 subjects compared with example 1, we obtain a power of roughly 76%.

Example 5: Multiple values of study parameters
In this example, we assess the effect of varying the value of the correlation of the male group on the

power of our study. We supply a list of correlations in parentheses for the second command argument:

. power twocorrelations 0.3 (0.4(0.1)0.9), n(500)
Estimated power for a two-sample correlations test
Fisher’s z test
H0: r2 = r1 versus Ha: r2 != r1

alpha power N N1 N2 delta r1 r2

.05 .2452 500 250 250 .1 .3 .4

.05 .7595 500 250 250 .2 .3 .5

.05 .9894 500 250 250 .3 .3 .6

.05 1 500 250 250 .4 .3 .7

.05 1 500 250 250 .5 .3 .8

.05 1 500 250 250 .6 .3 .9

From the table, we see that the power increases from 25% to 100% as the correlation increases from 0.4

to 0.9.

For multiple values of parameters, the results are automatically displayed in a table, as we see above.

For more examples of tables, see [PSS-2] power, table. If you wish to produce a power plot, see

[PSS-2] power, graph.

https://www.stata.com/manuals/pss-2powertwocorrelations.pdf#pss-2powertwocorrelationsRemarksandexamplesex1
https://www.stata.com/manuals/pss-2powertwocorrelations.pdf#pss-2powertwocorrelationsRemarksandexamplesex1
https://www.stata.com/manuals/pss-2powertable.pdf#pss-2power,table
https://www.stata.com/manuals/pss-2powergraph.pdf#pss-2power,graph

Computing effect size and experimental-group correlation
Effect size 𝛿 for a two-sample correlations test is defined as the difference between the experimental-

group and control-group correlations, 𝛿 = 𝜌2 − 𝜌1.

Sometimes, we may be interested in determining the smallest effect (for a given control-group cor-

relation) that yields a statistically significant result for prespecified sample size and power. In this case,

power, sample size, and control-group correlation must be specified. In addition, you must also decide

on the direction of the effect: upper, meaning 𝜌2 > 𝜌1, or lower, meaning 𝜌2 < 𝜌1. The direction may

be specified in the direction() option; direction(upper) is the default.

Example 6: Minimum detectable value of the experimental-group correlation
Continuing with example 4, we will compute the smallest positive correlation of the male group that

can be detected given a total sample of 500 individuals and a power of 80%. To solve for the experimental-

group correlation, after the command, we specify the control-group correlation of 0.3, total sample size
n(500), and power power(0.8):

. power twocorrelations 0.3, n(500) power(0.8)
Performing iteration ...
Estimated experimental-group correlation for a two-sample correlations test
Fisher’s z test
H0: r2 = r1 versus Ha: r2 != r1; r2 > r1
Study parameters:

alpha = 0.0500
power = 0.8000

N = 500
N per group = 250

r1 = 0.3000
Estimated effect size and experimental-group correlation:

delta = 0.2092
r2 = 0.5092

We find that the minimum value of the experimental-group correlation that would yield a statistically

significant result in this study is 0.5092, and the corresponding effect size is 0.2092.

In this example, we computed the correlation assuming an upper direction, 𝜌2 > 𝜌1, or a positive

effect, 𝛿 > 0. To request a lower direction, or a negative effect, we can specify the direction(lower)
option.

Testing a hypothesis about two independent correlations
After data are collected, we can use the mvtest command to test the equality of two independent

correlations using an asymptotic likelihood-ratio test; see [MV]mvtest for details. We can also manually

perform a two-sample Fisher’s 𝑧 test on which power twocorrelations bases its computations. We

demonstrate our examples using genderpsych.dta from [MV] mvtest correlations.

Example 7: Comparing two correlations using mvtest
Consider a sample of 64 individuals with equal numbers of males and females. Wewould like to know

whether the correlation between the pictorial inconsistencies (variable y1) and vocabulary (variable y4)
is the same between males and females.

https://www.stata.com/manuals/pss-2powertwocorrelations.pdf#pss-2powertwocorrelationsRemarksandexamplesex4
https://www.stata.com/manuals/mvmvtest.pdf#mvmvtest
https://www.stata.com/manuals/mvmvtestcorrelations.pdf#mvmvtestcorrelations

. use https://www.stata-press.com/data/r19/genderpsych
(Four psychological test scores, Rencher and Christensen (2012))
. mvtest correlations y1 y4, by(gender)
Test of equality of correlation matrices across samples

Jennrich chi2(1) = 2.16
Prob > chi2 = 0.1415

The reported 𝑝-value is 0.1415, so we do not have sufficient evidence to reject the null hypothesis about
the equality of the two correlations.

Example 8: Two-sample Fisher’s z test
To compute a two-sample Fisher’s 𝑧 test manually, we perform the following steps. We first compute

the estimates of correlation coefficients for each group by using the correlate command; see [R] cor-

relate. We then compute Fisher’s 𝑧 test statistic and its corresponding 𝑝-value.
We compute and store correlation estimates for males in the r1 scalar and the corresponding sample

size in the N1 scalar.

. /* compute and store correlation and sample size for males */

. correlate y1 y4 if gender==1
(obs=32)

y1 y4

y1 1.0000
y4 0.5647 1.0000

. scalar r1 = r(rho)

. scalar N1 = r(N)

We store the corresponding results for females in r2 and N2.

. /* compute and store correlation and sample size for females */

. correlate y1 y4 if gender==2
(obs=32)

y1 y4

y1 1.0000
y4 0.2596 1.0000

. scalar r2 = r(rho)

. scalar N2 = r(N)

We now compute the 𝑧 test statistic, stored in the Z scalar, by applying Fisher’s 𝑧 transformation to

the obtained correlation estimates r1 and r2 and using the cumulative function of the standard normal

distribution normal() to compute the 𝑝-value.
. /* compute Fisher’s z statistic and p-value and display results */
. scalar mu_Z = atanh(r2) - atanh(r1)
. scalar sigma_Z = sqrt(1/(N1-3)+1/(N2-3))
. scalar Z = mu_Z/sigma_Z
. scalar pvalue = 2*normal(-abs(Z))
. display ”Z statistic = ” %8.4g Z _n ”P-value = ” %8.4g pvalue
Z statistic = -1.424
P-value = .1543

The 𝑝-value is 0.1543 and is close to the 𝑝-value reported by mvtest in example 7.

https://www.stata.com/manuals/rcorrelate.pdf#rcorrelate
https://www.stata.com/manuals/rcorrelate.pdf#rcorrelate
https://www.stata.com/manuals/pss-2powertwocorrelations.pdf#pss-2powertwocorrelationsRemarksandexamplesex7

The estimates of the correlations obtained from correlate are 0.5647 for males and 0.2596 for

females. Wemay check howmany subjects we need to detect the difference between these two correlation

values.

. power twocorrelations 0.5647 0.2596
Performing iteration ...
Estimated sample sizes for a two-sample correlations test
Fisher’s z test
H0: r2 = r1 versus Ha: r2 != r1
Study parameters:

alpha = 0.0500
power = 0.8000
delta = -0.3051

r1 = 0.5647
r2 = 0.2596

Estimated sample sizes:
N = 232

N per group = 116

We need a total of 232 subjects, 116 per group in a balanced design, to have a power of 80% to detect

the difference between the given values of correlations for males and females.

Stored results
power twocorrelations stores the following in r():

Scalars

r(alpha) significance level

r(power) power

r(beta) probability of a type II error

r(delta) effect size

r(N) total sample size

r(N a) actual sample size

r(N1) sample size of the control group

r(N2) sample size of the experimental group

r(nratio) ratio of sample sizes, N2/N1
r(nratio a) actual ratio of sample sizes

r(nfractional) 1 if nfractional is specified, 0 otherwise

r(onesided) 1 for a one-sided test, 0 otherwise

r(r1) control-group correlation

r(r2) experimental-group correlation

r(diff) difference between the experimental- and control-group correlations

r(separator) number of lines between separator lines in the table

r(divider) 1 if divider is requested in the table, 0 otherwise

r(init) initial value for sample sizes or experimental-group correlation

r(maxiter) maximum number of iterations

r(iter) number of iterations performed

r(tolerance) requested parameter tolerance

r(deltax) final parameter tolerance achieved

r(ftolerance) requested distance of the objective function from zero

r(function) final distance of the objective function from zero

r(converged) 1 if iteration algorithm converged, 0 otherwise

Macros

r(type) test
r(method) twocorrelations
r(direction) upper or lower
r(columns) displayed table columns

r(labels) table column labels

r(widths) table column widths

r(formats) table column formats

Matrices

r(pss table) table of results

Methods and formulas
Let 𝜌1 and 𝜌2 denote Pearson’s correlation for the control and the experimental groups, respectively.

Let ̂𝜌1 and ̂𝜌2 denote the corresponding estimators, the sample correlation coefficients.

A two-sample correlations test involves testing the null hypothesis 𝐻0∶ 𝜌2 = 𝜌1 versus the two-sided

alternative hypothesis 𝐻𝑎∶ 𝜌2 ≠ 𝜌1, the upper one-sided alternative 𝐻𝑎∶ 𝜌2 > 𝜌1, or the lower one-sided

alternative 𝐻𝑎∶ 𝜌2 < 𝜌1.

The exact distribution of the difference between the sample correlation coefficients ̂𝜌1 and ̂𝜌2 is com-

plicated for testing the hypothesis 𝐻0 ∶ 𝜌1 − 𝜌2 ≠ 0. An alternative is to apply an inverse hyperbolic

tangent, tanh−1(⋅), or Fisher’s 𝑧 transformation to the estimators to approximate their sampling dis-

tribution by a normal distribution (Fisher 1915). The hypothesis 𝐻0 ∶ 𝜌1 = 𝜌2 is then equivalent to

𝐻0∶ tanh
−1(𝜌1) = tanh−1(𝜌2).

Let 𝑍1 = tanh−1( ̂𝜌1) = 0.5 ln{(1 + ̂𝜌1)/(1 − ̂𝜌1)} denote Fisher’s 𝑧 transformation of ̂𝜌1, and

let 𝑍2 denote the corresponding transformation of ̂𝜌2. Then the difference 𝛿𝑧 = 𝑍2 − 𝑍1 follows a

normal distribution with mean 𝜇𝑧 = 𝜇2 − 𝜇1 = tanh−1(𝜌2) − tanh−1(𝜌1) and standard deviation 𝜎𝑧 =
√1/(𝑛1 − 3) + 1/(𝑛2 − 3), where 𝑛1 and 𝑛2 are the sample sizes of the control and the experimental

groups, respectively (for example, see Graybill [1961, 211] and Anderson [2003, 134]).

Let 𝛼 be the significance level, 𝛽 be the probability of a type II error, and 𝑧1−𝛼 and 𝑧𝛽 be the (1−𝛼)th
and the 𝛽th quantiles of the standard normal distribution.

The power 𝜋 = 1 − 𝛽 is computed using

𝜋 =
⎧{
⎨{⎩

Φ ( 𝛿𝑧
𝜎𝑧

− 𝑧1−𝛼) for an upper one-sided test

Φ (− 𝛿𝑧
𝜎𝑧

− 𝑧1−𝛼) for a lower one-sided test

Φ ( 𝛿𝑧
𝜎𝑧

− 𝑧1−𝛼/2) + Φ (− 𝛿𝑧
𝜎𝑧

− 𝑧1−𝛼/2) for a two-sided test

(1)

where Φ(⋅) is the cdf of a standard normal distribution.
Let 𝑅 = 𝑛2/𝑛1 denote the allocation ratio. Then 𝑛2 = 𝑅 × 𝑛1 and power can be viewed as a

function of 𝑛1. Therefore, for sample-size determination, the control-group sample size 𝑛1 is computed

first. The experimental-group size 𝑛2 is then computed as 𝑅 × 𝑛1, and the total sample size is computed

as 𝑛 = 𝑛1 + 𝑛2. By default, sample sizes are rounded to integer values; see Fractional sample sizes in

[PSS-4] Unbalanced designs for details.

For a one-sided test, the control-group sample size 𝑛1 is computed as a positive root of the following

quadratic equation:

(𝑅 + 1)𝑛1 − 6
(𝑛1 − 3)(𝑅𝑛1 − 3)

= ( 𝛿𝑧
𝑧1−𝛼 − 𝑧𝛽

)
2

https://www.stata.com/manuals/pss-4unbalanceddesigns.pdf#pss-4UnbalanceddesignsRemarksandexamplesFractionalsamplesizes
https://www.stata.com/manuals/pss-4unbalanceddesigns.pdf#pss-4Unbalanceddesigns

For a one-sided test, if one of the group sizes is known, the other one is computed using the following

formula. For example, to compute 𝑛1 given 𝑛2 we use

𝑛1 = 3 + 1

( 𝛿𝑧
𝑧1−𝛼−𝑧𝛽

)
2

− 1
𝑛2−3

The minimum detectable value of the experimental-group correlation is obtained by first finding the

corresponding minimum value of 𝜇2 and then applying the inverse Fisher’s 𝑧 transformation to that 𝜇2:

𝜌2 = 𝑒2𝜇2 − 1
𝑒2𝜇2 + 1

For a one-sided test, 𝜇2 = 𝜇1 + 𝜎𝑧(𝑧1−𝛼 − 𝑧𝛽) when 𝜇2 > 𝜇1, and 𝜇2 = 𝜇1 − 𝜎𝑧(𝑧1−𝛼 − 𝑧𝛽) when
𝜇2 < 𝜇1.

For a two-sided test, the sample sizes and 𝜇2 are obtained by iteratively solving the two-sided power

equation in (1) for the respective parameters. The initial values are obtained from the respective formulas

for the one-sided computation with the significance level 𝛼/2.

